

[Click Here](#) to access the data file associated with these examples

```
*-----;
* EX01: make data, vanilla REPORT
*-----;
data survey;
  infile 'c:\procprep\procprep.dat';

  input @01 fname      $char10.
        @12 linit     $char1.
        @15 sex       $char1.
        @18 acadyear  $char1.
        @21 gpa       3.1
        @27 age       2.
        @31 books     4.
        @40 food      4.
        @49 ent       4.;

run;

*-----;
proc format;
  value $sexfmt  'M' = 'MALE'
                'F' = 'FEMALE';

  value $yearfmt '1' = 'FRESHMAN'
                '2' = 'SOPHOMORE'
                '3' = 'JUNIOR'
                '4' = 'SENIOR';

run;

*-----;
proc report nowd data=survey;
run;

*-----;

*-----;
* EX02: basic statements
*-----;
proc report nowd data=survey;

  column fname linit sex acadyear gpa age books food ent;

  define sex      / display      format=$sexfmt.  width=6;
  define acadyear / display 'ACADEMIC/YEAR' format=$yearfmt. width=9;
  define gpa      / analysis    format=3.1          width=3;
  define age      / analysis    format=3.          width=3;
  define books    / analysis '$-BOOKS' format=dollar6.  width=7;
  define food     / analysis '$-FOOD' format=dollar6.  width=6;
  define ent      / analysis '$-ENTERTAIN' format=dollar6.  width=11;

run;

*-----;
```

```

*-----;
* EX03: COMPUTED variables
*-----;
proc report nowd data=survey;

  column fname linit name sex acadyear gpa age books food ent total;

  define sex / display format=$sexfmt. width=6;
  define acadyear / display 'ACADEMIC/YEAR' format=$yearfmt. width=9;
  define gpa / analysis format=3.1 width=3;
  define age / analysis format=3. width=3;
  define books / analysis '$-BOOKS' format=dollar6. width=7;
  define food / analysis '$-FOOD' format=dollar6. width=6;
  define ent / analysis '$-ENTERTAIN' format=dollar6. width=11;
  define name / computed width=10;
  define total / computed '$-TOTAL' format=dollar6. width=7;

  compute name / char length=10;
  name = trim(fname) || ' ' || linit || '.';
endcomp;

  compute total;
  total = sum(books.sum, food.sum, ent.sum);
endcomp;
run;

```

```

*-----;
* EX04: NOPRINT
*-----;
proc report nowd data=survey;

  column fname linit name sex acadyear gpa age books food ent total;

  define fname / display noprint;
  define linit / display noprint;
  define sex / display format=$sexfmt. width=6;
  define acadyear / display 'ACADEMIC/YEAR' format=$yearfmt. width=9;
  define gpa / analysis format=3.1 width=3;
  define age / analysis format=3. width=3;
  define books / analysis '$-BOOKS' format=dollar6. width=7;
  define food / analysis '$-FOOD' format=dollar6. width=6;
  define ent / analysis '$-ENTERTAIN' format=dollar6. width=11;
  define name / computed width=10;
  define total / computed '$-TOTAL' format=dollar6. width=7;

  compute name / char length=10;
  name = trim(fname) || ' ' || linit || '.';

```

```

endcomp;

compute total;
  total = sum(books.sum, food.sum, ent.sum);
endcomp;
run;
*-----;

```

```

*-----;
* EX05: HEADLINE, HEADSKIP
*-----;

```

```

proc report nowd data=survey headline headskip;

  column fname linit name sex acadyear gpa age books food ent total;

  define fname      / display  noprint;
  define linit      / display  noprint;
  define sex        / display                format=$sexfmt.   width=6;
  define acadyear   / display  'ACADEMIC/YEAR' format=$yearfmt. width=9;
  define gpa        / analysis                format=3.1       width=3;
  define age        / analysis                format=3.         width=3;
  define books      / analysis '$-BOOKS'      format=dollar6.   width=7;
  define food       / analysis '$-FOOD'       format=dollar6.   width=6;
  define ent        / analysis '$-ENTERTAIN'   format=dollar6.   width=11;
  define name       / computed                width=10;
  define total      / computed '$-TOTAL'      format=dollar6.   width=7;

  compute name / char length=10;
    name = trim(fname) || ' ' || linit || '.';
  endcomp;

  compute total;
    total = sum(books.sum, food.sum, ent.sum);
  endcomp;
run;
*-----;

```

```

*-----;
* EX06: switch order of a few variables
*-----;

```

```

proc report nowd data=survey headline headskip;

  column acadyear sex fname linit name gpa age books food ent total;

  define fname      / display  noprint;
  define linit      / display  noprint;
  define sex        / display                format=$sexfmt.   width=6;
  define acadyear   / display  'ACADEMIC/YEAR' format=$yearfmt. width=9;
  define gpa        / analysis                format=3.1       width=3;

```

```

define age      / analysis      format=3.      width=3;
define books   / analysis '$-BOOKS' format=dollar6. width=7;
define food    / analysis '$-FOOD' format=dollar6. width=6;
define ent     / analysis '$-ENTERTAIN' format=dollar6. width=11;
define name    / computed      width=10;
define total   / computed '$-TOTAL' format=dollar6. width=7;

```

```

compute name / char length=10;
name = trim(fname) || ' ' || linit || '.';
endcomp;

```

```

compute total;
total = sum(books.sum, food.sum, ent.sum);
endcomp;

```

```
run;
```

```
*-----;
```

```
*-----;
```

```
* EX07: ORDER variables ;
```

```
*-----;
```

```
proc report nowd data=survey headline headskip;
```

```
column acadyear sex fname linit name gpa age books food ent total;
```

```

define fname    / display noprint;
define linit    / display noprint;
define sex      / order      format=$sexfmt. width=6;
define acadyear / order      'ACADEMIC/YEAR' format=$yearfmt. width=9

```

```
order=internal;
```

```

define gpa      / analysis      format=3.1      width=3;
define age      / analysis      format=3.      width=3;
define books    / analysis '$-BOOKS' format=dollar6. width=7;
define food     / analysis '$-FOOD' format=dollar6. width=6;
define ent      / analysis '$-ENTERTAIN' format=dollar6. width=11;
define name     / computed      width=10;
define total    / computed '$-TOTAL' format=dollar6. width=7;

```

```

compute name / char length=10;
name = trim(fname) || ' ' || linit || '.';
endcomp;

```

```

compute total;
total = sum(books.sum, food.sum, ent.sum);
endcomp;

```

```
run;
```

```
*-----;
```

```
*-----;
```

```
* EX08: GROUP variables ;
```

```

*-----;
proc report nowd data=survey headline headskip;

    column acadyear sex gpa age books food ent total;

    define sex / group format=$sexfmt. width=6;
    define acadyear / group 'ACADEMIC/YEAR' format=$yearfmt. width=9
order=internal;
    define gpa / analysis format=3.1 width=3;
    define age / analysis format=3. width=3;
    define books / analysis '$-BOOKS' format=dollar6. width=7;
    define food / analysis '$-FOOD' format=dollar6. width=6;
    define ent / analysis '$-ENTERTAIN' format=dollar6. width=11;
    define total / computed '$-TOTAL' format=dollar6. width=7;

    compute total;
        total = sum(books.sum, food.sum, ent.sum);
    endcomp;
run;
*-----;

```

```

*-----;
* EX09: MEAN statistic ;
*-----;
proc report nowd data=survey headline headskip;

    column acadyear sex gpa age books food ent total;

    define sex / group format=$sexfmt. width=6;
    define acadyear / group 'ACADEMIC/YEAR' format=$yearfmt. width=9
order=internal;
    define gpa / analysis 'MEAN/GPA' format=3.1 width=4
mean;
    define age / analysis 'MEAN/AGE' format=3. width=4
mean;
    define books / analysis '$-BOOKS' format=dollar6. width=7
mean;
    define food / analysis '$-FOOD' format=dollar7. width=7
mean;
    define ent / analysis '$-ENTERTAIN' format=dollar7. width=11
mean;
    define total / computed '$-TOTAL' format=dollar7. width=7;

    compute total;
        total = sum(books.mean, food.mean, ent.mean);
    endcomp;
run;
*-----;

```

```

*-----;
* EX10: column spanning headers
*-----;
title1 'PROC REPORT - SAMPLE SUMMARY REPORT';
title2 '----- ACADEMIC SURVEY -----';
title3;
*-----;
proc report nowd data=survey headline headskip ps=30;

    column acadyear sex gpa age ('-ANNUAL MEAN AMOUNT SPENT ON-' books food
ent total);

    define sex      / group      format=$sexfmt.  width=6;
    define acadyear / group      'ACADEMIC/YEAR' format=$yearfmt. width=9
order=internal;
    define gpa      / analysis 'MEAN/GPA'          format=3.1      width=4
mean;
    define age      / analysis 'MEAN/AGE'          format=3.        width=4
mean;
    define books    / analysis '$-BOOKS'          format=dollar6. width=7
mean;
    define food     / analysis '$-FOOD'           format=dollar7. width=7
mean;
    define ent      / analysis '$-ENTERTAIN'      format=dollar7. width=11
mean;
    define total    / computed '$-TOTAL'         format=dollar7. width=7;

    compute total;
        total = sum(books.mean, food.mean, ent.mean);
    endcomp;
run;
*-----;

*-----;
* EX11: BREAKs, aliases, alternate underlines
*-----;
title1 'PROC REPORT - SAMPLE SUMMARY REPORT';
title2 '----- ACADEMIC SURVEY -----';
title3;
*-----;
proc report nowd data=survey split='\';

    column acadyear sex gpa age ('-ANNUAL AMOUNT SPENT ON-'
                                ('-BOOKS-'      books books=booksm)
                                ('-FOOD-'       food  food =foodm)
                                ('-ENTERTAIN-'  ent   ent  =entm)
                                ('-TOTAL-'      total totalm) );

    define sex      / group      'SEX\--'          format=$sexfmt.
width=6;

```

```

    define acadyear / group 'ACADEMIC\YEAR\--' format=$yearfmt. width=9
order=internal;
    define gpa / analysis 'MEAN\GPA\--' format=3.1 width=4
mean;
    define age / analysis 'MEAN\AGE\--' format=3. width=4
mean;
    define books / analysis 'TOTAL\--' format=dollar7. width=7
sum;
    define booksm / analysis 'MEAN\--' format=dollar4. width=4
mean;
    define food / analysis 'TOTAL\--' format=dollar7. width=7
sum;
    define foodm / analysis 'MEAN\--' format=dollar4. width=4
mean;
    define ent / analysis 'TOTAL\--' format=dollar7. width=7
sum;
    define entm / analysis 'MEAN\--' format=dollar4. width=4
mean;
    define total / computed 'TOTAL\--' format=dollar8.
width=8;
    define totalm / computed 'MEAN\--' format=dollar7.
width=7;

    compute total;
        total = sum(books.sum, food.sum, ent.sum);
    endcomp;

    compute totalm;
        totalm = sum(booksm, foodm, entm);
    endcomp;

    break after acadyear / ol ul skip summarize suppress;

    rbreak after / dol dul skip summarize;
run;
*-----;

*-----;
* EX12: PANELS ;
*-----;
title1 'PROC REPORT - SAMPLE SUMMARY REPORT';
title2 '----- ACADEMIC SURVEY -----';
title3;
*-----;
proc report nowd data=survey headline headskip
panels=3 ps=50 pspace=10 ls=110;

column fname linit name acadyear gpa;

define fname / display noprint;

```

```

define linit / display noprint;
define acadyear / display 'ACADEMIC/YEAR' format=$yearfmt. width=9;
define gpa / analysis format=3.1 width=3;
define name / computed width=10;

compute name / char length=10;
name = trim(fname) || ' ' || linit || '.';
endcomp;
run;
*-----;

```

```

*-----;
* EX13: FLOW ;
*-----;

```

```

title1 'PROC REPORT - SAMPLE SUMMARY REPORT';
title2 '----- ACADEMIC SURVEY -----';
title3;

```

```

*-----;

```

```

proc report nowd data=survey split='\ '
panels=99 ps=25 ls=150;

```

```

column fname linit acadyear age sex gpa block;

```

```

define fname / display noprint;
define linit / display noprint;
define acadyear / display noprint;
define age / display noprint;
define sex / display noprint;
define gpa / display noprint;
define block / computed ' ' flow width=20 '';

```

```

compute block / char length=100;
block= 'NAME: ' || trim(fname) || ' ' || linit || '.'
|| '\YEAR: ' || put(acadyear,$yearfmt.)
|| '\ AGE: ' || put(age,2.)
|| '\ SEX: ' || put(sex,$sexfmt.)
|| '\ GPA: ' || put(gpa,3.1)
|| '\\';
endcomp;

```

```

run;
*-----;

```

```

*-----;
* EX14: WRAP ;
*-----;

```

```

title1 'PROC REPORT - SAMPLE SUMMARY REPORT';
title2 '----- ACADEMIC SURVEY -----';
title3;

```

```

*-----;

```

```

data survey2;
  set survey;
  obsno = _n_;
run;
*-----;
proc report nowd data=survey2 headline headskip
  named wrap ls=70;

  column obsno fname linit name acadyear age sex gpa books food ent total;

  define obsno / order noprint;
  define fname / display noprint;
  define linit / display noprint;
  define sex / display format=$sexfmt. width=6;
  define acadyear / display 'YEAR' format=$yearfmt. width=9;
  define gpa / analysis format=3.1 width=3;
  define age / analysis format=3. width=3;
  define books / analysis '$-BOOKS' format=dollar6. width=6;
  define food / analysis '$-FOOD' format=dollar6. width=6
spacing=3;
  define ent / analysis '$-ENTERTAIN' format=dollar6. width=6
spacing=3;
  define name / computed width=10;
  define total / computed '$-TOTAL' format=dollar6. width=7
spacing=3;

  compute name / char length=10;
  name = trim(fname) || ' ' || linit || '.';
endcomp;

  compute total;
  total = sum(books.sum, food.sum, ent.sum);
endcomp;

  break after obsno / skip;
run;
*-----;

*-----;
* EX15: ACROSS variables, output data set ;
*-----;
title1 'PROC REPORT - SAMPLE SUMMARY REPORT';
title2 '----- ACADEMIC SURVEY -----';
title3;
*-----;
proc report nowd data=survey headline headskip out=out_ex;

  column acadyear sex,gpa;

  define sex / across format=$sexfmt. width=6;

```

```
    define acadyear / group      'ACADEMIC/YEAR' format=$yearfmt. width=9
order=internal;
    define gpa      / analysis      format=3.1      width=6
mean;
run;
*-----;
title1 "CONTENTS OF PROC REPORT OUTPUT DATA SET";
title2 "----- PRODUCED WITH 'OUT=' OPTION -----";
title3;-----;
proc print data=out_ex;
run;
*-----;
```